

Section

02

オートSUMで関数を入力しよう

ここで学ぶこと

- ・オートSUM
- ・合計
- ・SUM関数

数値を合計するには、[オートSUM]をクリックするだけでSUM関数が入力された状態になります。自動的に認識された合計範囲が合っていれば、あとは[Enter]を押すだけです。簡単、手間なしで数値を合計できます。

練習▶02_営業成績

解説

[オートSUM]を使う

ここでは、[オートSUM]を利用して営業員別の売上を合計します。[オートSUM]には、SUM関数が割り当てられています。クリックするだけでSUM関数が入力されます。また、SUM関数以外の代表的な4つの関数も入力することができます(右ページ補足参照)。


ヒント

[ホーム]タブでも[オートSUM]が利用できる


[オートSUM]は[ホーム]タブにも[合計]として用意されています。どちらを使っても同じです。[ホーム]タブで表の編集などを行っているときは、[ホーム]タブの[合計]を使うと、[数式]タブに切り替える手間が省けます。

1 [オートSUM]を使って関数を入力する

1 関数を入力したいセルをクリックし、



2 [数式]タブの[オートSUM]をクリックすると、




3 合計を求めるSUM関数が自動的に入力されます。

ショートカットキー


[Ctrl] + [Enter] でも関数を確定できる

[Enter]を押すと、関数が確定すると同時にアクティブセルが下に移動しますが、[Ctrl]を押しながら[Enter]を押すとアクティブセルを移動せずに確定できます。

4 引数を指定します。ここでは、自動的に認識されたセル範囲の内容を確認します。



5 [Enter]を押すと、関数の入力が確定します。




補足

[オートSUM]から入力できる関数

[オートSUM]で入力できる関数は、[合計]、[平均]、[数値の個数]、[最大値]、[最小値]の5種類です。関数を入力するセルが、数値の入ったセルに隣接している場合は、26ページの手順3のように、引数が自動的に認識されます。ただし、自動的に認識されるセルやセル範囲はいつも正しいとは限りません。誤っている場合は、27ページの手順4のタイミングで、正しいセルやセル範囲をドラッグなどで指定し直します。

1 [オートSUM]の右の[]をクリックすると、



2 代表的な関数の一覧が表示されます。

合計しよう

ここで学ぶこと

- SUM
- 3-D参照
- 串刺し集計

売上や費用の合計など、**数値を合計する**には**SUM関数**を使います。いわゆる足し算をする関数ですが、表の所々にある小計を自動的に見分けて合計したり、シートをまたがって合計したりすることができます。

練習 ▶ 07_経費管理表、07_売上分析表、07_3店集計

書式

SUM(数値1[,数値2]…)

重要用語

SUM

SUM関数は、引数に指定した数値やセル、セル範囲の合計値を求めます。

引数

数値

数値や数値の入ったセル、セル範囲を指定します。合計したい数値のセルが離れている場合は、「,(カンマ)」で区切りながら指定します。セルやセル範囲に含まれる空白セルや文字列、論理値(80ページ参照)は無視されます。

1 総計を求める

SUM

解説

総計を求める

ここでは、水道光熱費と通信費の小計を合計し、経費の合計を求めます。

ヒント

小計は総計の前に数式かSUM関数で求めておく

小計は総計より先に足し算の数式かSUM関数で求めておきます。小計のセルが空白、または、数値の場合は、小計と認識できないためです。自動認識できない場合には、カンマで区切りながら、小計のセルを指定します。

=SUM(C11,C6)

1	2	3	4	5	6	7	8	9	10	11	12
1	経費管理表										
2	科目	費目	4月	5月	6月						
3	水道光熱費	電気	10,000	12,800	13,500						
4		水道	3,000		2,850						
5		ガス	2,000	3,200	2,650						
6		小計	15,000	16,000	19,000						
7	通信費	電話・インターネット	13,000	11,500	12,800						
8		はがき	6,000	0	2,400						
9		切手	400	40	820						
10		郵送費	17,000	15,500	20,800						
11		小計	36,400	27,040	36,820						
12		総計	51,400								

- 1 総計のセル[C12]をクリックし、[ホーム]タブまたは[数式]タブの[オートSUM]をクリックすると、小計のセル[C6]と[C11]を自動的に認識して合計します。

2 売上累計を求める

SUM

解説

累計を求める

ここでは、部門別売上高から売上高と構成比を累計します。累計は、数値を順次合計した値です。最初の数値からスタートし、最初の数値に2番目の数値を足して2番目までの累計を求めています。続いて、最初から2番目までの累計に3番目の数値を足して3番目までの累計を求めます。表の末尾まで繰り返すと、総計に一致します(下のグラフ参照)。

=SUM(B\$3:B3)


=SUM(C\$3:C3)

D3	A	B	C	D	E	F
1	売上分析表					
2	部門	合計売上高	売上構成比	累計売上高	累計構成比	
3	衣料品	109,830	36.1%	109,830	36.1%	
4	食料品	88,400	29.0%	198,230	65.1%	
5	雑貨	43,350	14.2%	241,580	79.3%	
6	身の回り品	38,490	12.6%	280,070	91.9%	
7	家庭用品	15,750	5.2%	295,820	97.1%	
8	食堂	8,780	2.9%	304,600	100.0%	
9	総計	304,600				
10						

- 1 引数の[数値1]にセル範囲[B\$3:B3]と指定し、オートフィルで下方向にコピーしたときに、セル範囲の先頭を絶対参照で固定したまま、末尾のセルが1行ずつ下に移動し、セル範囲が拡張するようにします。
- 2 売上構成比の累計も同様です。セル範囲[C\$3:C3]と指定します。

ヒント 累計と総計の関係

下のグラフに示すように、累計は、1つずつ数値が積み上がり、表の末尾まで累計すると総計に一致します。累計に使う数値は、最初の数値から2番目まで、3番目までというように、1つずつ拡張していることもわかります。



- 1 衣料品から食料品までの累計です。
- 2 雑貨を足して、衣料品から雑貨までの累計になります。
- 3 数値が1つずつ積み上がるのは、セル範囲が1つずつ拡張するのと同じです。

Section

15

平均を求めよう

ここで学ぶこと

- AVERAGE
- AVERAGEA
- 平均

一般に、平均値は個々の数値を全部足して数値の個数で割った値です。数値の**平均値を求める**には**AVERAGE関数**を利用しますが、何を基準に平均するのかによっては、AVERAGE関数は利用できない場合があります。

練習▶15_バター売上表、15_バター売上表_02

書式 **AVERAGE(数値1[,数値2]…)**
AVERAGEA(値1[,値2]…)

重要用語

AVERAGE / AVERAGEA

AVERAGE関数は、指定したセル範囲にある数値を計算対象とし、数値の平均値を求めます。また、AVERAGEA関数は、空白以外の値の平均値を求めます。なお、AVERAGE関数は、[オートSUM]の[平均]でも入力できます(27ページ参照)。

解説

値の種類が混在する場合の戻り値の違い

右の図のセル範囲[D2:D7]のうち、AVERAGE関数の計算対象は、セル[D2]の1のみです。よって、平均値は数値「1」÷数値の数「1」=1です。AVERAGEA関数は、セル[D7]の空白セル以外の5個が計算対象です。このうち、数値と論理値「TRUE」が1で合計2、残りは0です。よって、平均値は2÷5=0.4になります。

引数

数値

数値、または、数値のセルやセル範囲を指定します。

値

任意の値、セルやセル範囲を指定します。

AVERAGE関数とAVERAGEA関数の計算対象

データが数値だけの場合は、AVERAGE/AVERAGEA関数とも同じです。データに複数の値の種類が混在する場合、AVERAGE関数は、数値のみを計算対象とします。AVERAGEA関数では、文字列、[Space]を押して入力した空白文字、FALSEを「0」、TRUEを「1」として計算します。両関数とも空白セルは無視します。

=AVERAGE(D2:D7)			
値の種類	値	値の種類	値
数値	1	数値	1
数値	3	空白文字	
数値	4	文字列	abc
数値	5	論理値	FALSE
数値	6	論理値	TRUE
数値	2	空白セル	
AVERAGE関数	3.5	AVERAGE関数	1.0
AVERAGEA関数	3.5	AVERAGEA関数	0.4

1 売上日あたりの平均販売価格を求める

AVERAGE

解説

1行あたりの平均はAVERAGE関数を使う

ここでは、バターの売上日あたりの平均販売価格を求めます。この例では、日付単位で売上データが入力されています。つまり、売上日あたりの平均販売価格は、1件あたりの平均値であり、AVERAGE関数で求めます。

=AVERAGE(C3:C154)

▼4月-9月 バター売上表					▼売上分析		
日付	商品	販売価格	数量	売上金額	◎バターの売上日あたりの平均販売価格		
2022/4/1	バター	368	33	12,144	平均販売価格/日		
2022/4/2	バター	448	2	896			
2022/4/5	バター	368	33	12,144	◎バターの売上1個あたりの平均販売価格		
2022/4/6	バター	448	9	4,032	平均販売価格/個		
2022/4/7	バター	288	50	14,400			
2022/4/8	バター	288	50	14,400			
2022/4/9	バター	288	50	14,400			
2022/4/11	バター	368	24	8,832			
2022/4/12	バター	368	32	11,776			
2022/4/13	バター	448	6	2,688			

- ①「販売価格」のセル範囲[C3:C154]を[数値]に指定します。「販売価格」の列に「販売価格」に無関係な数値が入らない場合は、列番号[C]をクリックして、C列全体を[数値]に指定することもできます。

2 売上1個あたりの平均販売価格を求める

SUM

解説

AVERAGE関数が利用できない平均

ここでは、バターの売上1個あたりの平均販売価格を求めます。バター売上表は、日付単位の表のため、数量1個あたりの平均にAVERAGE関数を使うことはできません。そこで、「販売価格=売上金額÷数量」の関係から、売上金額の合計を数量の合計で割って、1個あたりの販売価格を求めています。

=SUM(E3:E154)/SUM(D3:D154)

▼4月-9月 バター売上表					▼売上分析		
日付	商品	販売価格	数量	売上金額	◎バターの売上日あたりの平均販売価格		
2022/4/1	バター	368	33	12,144	平均販売価格/日		
2022/4/2	バター	448	2	896			
2022/4/5	バター	368	33	12,144	◎バターの売上1個あたりの平均販売価格		
2022/4/6	バター	448	9	4,032	平均販売価格/個		
2022/4/7	バター	288	50	14,400			
2022/4/8	バター	288	50	14,400			
2022/4/9	バター	288	50	14,400			
2022/4/11	バター	368	24	8,832			
2022/4/12	バター	368	32	11,776			
2022/4/13	バター	448	6	2,688			

- ①「売上金額」のセル範囲[E3:E154]をSUM関数の[数値]に指定し、売上金額の合計を求めます。
- ②「数量」のセル範囲[D3:D154]をSUM関数の[数値]に指定し、数量の合計を求めます。①を②で割って、1個あたりの平均販売価格を求めています。

Section

19

表示されたデータを 集計しよう

ここで学ぶこと

- SUBTOTAL
- フィルター
- 集計方法

一覧表形式の表にフィルターを設定すると、設定した条件に合うデータ行が抽出され、条件に合わないデータ行は折りたたまれて非表示になります。**非表示行を除外して**、データを集計したい場合は、**SUBTOTAL 関数**を使います。

 練習▶19_合計金額

書式

SUBTOTAL(集計方法,参照1 [,参照2]…)

 重要用語

SUBTOTAL

指定したセル範囲の値を、指定した集計方法で集計します。集計方法は11種類あります。フィルター機能による抽出で行が非表示になる場合は、「1～11」または「101～111」のどちらの[集計方法]を指定しても同様の結果になります。ただし、行番号を指定して[非表示]にした場合は、指定する[集計方法]によって関数の動作が変わります(75ページ解説参照)。

引 数

集計方法

集計内容に対応する番号を指定します（以下の表1を参照）。

参照

集計対象のセル範囲を指定します。

■ 表1 集計方法

集計方法 非表示行を含めて集計		集計内容	対応関数
する	しない		
1	101	平均	AVERAGE
2	102	個数	COUNT
3	103	空白以外の個数	COUNTA
4	104	最大値	MAX
5	105	最小値	MIN
6	106	積（掛け算）	PRODUCT
7	107	標本標準偏差	STDEV.S
8	108	標準偏差	STDEV.P
9	109	合計	SUM
10	110	不偏分散	VAR.S
11	111	分散	VAR.P

表に記載されている「非表示行を含めて集計」の「非表示行」とは、行番号を指定して非表示にした場合です。

① フィルターに応じて合計を求める

SUBTOTAL

 解説

フィルターを設定する

ここでは、抽出した数値を合計するため、
フィルターを準備します。

 解説

表の選択方法

フィルターを設定する場合、表の中の任意のセルを1箇所だけ選択すれば、Excelが表の範囲を自動認識しますが、正しい範囲を認識するとは限りません。右図の方法以外に、マウスを使う選択方法もあります。表の左上角(先頭)のセルをクリックし、表の末尾までスクロールで移動して表の右下隅(末尾)のセルで **[Shift]** を押しながらクリックします。

1 先頭のセル [A3] をクリックし、**[Shift]** と **[Ctrl]** を押しながら **[→]** を押して、表の右端まで選択します。

	A	B	C	D	E	F	G
1	合計金額						
2							
3	日付	費目	摘要	支払先等	金額		
4	2018/11/2	図書費	書籍代	技評ブック	1,980		
5	2018/11/2	旅費交通費	ガソリン代		5,230		

2 と を押しながら を押して、表の下端まで選択します。

3 [データ] タブの [フィルター] をクリックします。

4 フィルターが設定され、各列項目にフィルターボタンが表示されました。

	A	B	C	D	E	F	G
1	合計金額						
2							
3	日付	費目	摘要	支払先等	金額		
4	2018/11/2	図書費	書籍代	技研ブック	1,980		
5	2018/11/2	旅費交通費	ガソリン代		5,230		
6	2018/11/2	光熱費	ガス代		7,260		
7	2018/11/2	通信費	郵便代		680		
8	2018/11/4	光熱費	電気代		15,960		

23

複数の条件を判定しよう

ここで学ぶこと

- AND
- OR
- 論理式

データに複数の条件を付ける場合、**すべての条件を満たすAND条件**と**いずれか1つの条件を満たすOR条件**の2通りの条件の付け方があります。それぞれ**AND関数**と**OR関数**を利用します。

練習 ▶ 23_店舗別評価、23_店舗別評価_02、23_店舗別評価_03、23_店舗別評価_04、23_店舗別評価_05

書式

AND(論理式1[,論理式2]…)
OR(論理式1[,論理式2]…)

重要用語

AND / OR

AND関数は、[論理式]に指定した条件をすべて満たす場合に「TRUE」を、1つでも満たさない場合は「FALSE」を表示します。OR関数は、1つでも条件を満たせば「TRUE」を表示し、すべての条件に合わない場合のみ「FALSE」を表示します。


引数

論理式

条件式を指定します。条件式は、比較演算子を使った比較式や比較式の入ったセル、または、論理値を返す数式や関数を指定します。

AND関数とOR関数の戻り値

AND関数とOR関数の結果は、論理値「TRUE」、または、「FALSE」のいずれかです。以下の図のように、AND関数はAND条件、OR関数はOR条件での判定となります。なお、論理値「TRUE」は「真」、「条件が成立している」、論理値「FALSE」は「偽」、「条件が成立しない」という言い方もします。



① すべての評価が平均以上かどうか判定する

AND

解説

AND条件を判定する

ここでは、すべての評価が、全店舗の平均評価以上かどうかを店舗別に判定します。各評価が平均評価以上どうかを調べ、すべての評価で平均以上のときに「TRUE」になります。1つでも評価が平均を下回る場合は、「FALSE」が表示されます。

=AND(B3>=\$B\$10,C3>=\$C\$10,D3>=\$D\$10)

	A	B	C	D	E	F	G	H
1	店舗別評価一覧							
2	店舗名	接客	レジ待ち	品揃え	判定			
3	中野駅前	1	3	3	FALSE			
4	阿佐ヶ谷南	5	5	4	TRUE			
5	三鷹	3	1	4	FALSE			
6	武蔵境	2	2	3	FALSE			
7	小金井	1	3	1	FALSE			
8	立川北口	4	5	4	TRUE			
9	拝島南	4	1	5	FALSE			
10	平均評価	2.9	2.9	3.4				
11								

- ① 店舗の接客評価と接客の平均評価を比較する「B3>=\$B\$10」を[論理式1]に指定し、接客が平均評価以上かどうかを判定します。
- ② [論理式2]に「C3>=\$C\$10」と指定し、「レジ待ち」の評価を判定します。
- ③ [論理式3]に「D3>=\$D\$10」と指定し、「品揃え」の評価を判定します。

② 1つでも平均以上があるかどうか判定する

OR

解説

OR条件を判定する

ここでは、各評価のうち、1つでも平均評価以上があるかどうか判定します。AND条件とOR条件は、条件に対する判定方法が異なるだけで、個々の条件は同じです。したがって、引数を変更せずに関数名を入れ替えることで、判定方法を変えることができます。

=OR(B3>=\$B\$10,C3>=\$C\$10,D3>=\$D\$10)

	A	B	C	D	E	F	G	H
1	店舗別評価一覧							
2	店舗名	接客	レジ待ち	品揃え	判定			
3	中野駅前	1	3	3	TRUE			
4	阿佐ヶ谷南	5	5	4	TRUE			
5	三鷹	3	1	4	TRUE			
6	武蔵境	2	2	3	FALSE			
7	小金井	1	3	1	TRUE			
8	立川北口	4	5	4	TRUE			
9	拝島南	4	1	5	TRUE			
10	平均評価	2.9	2.9	3.4				
11								

- ① 利用例①の関数名を「AND」から「OR」に変更します。
- ② 引数はAND関数と同じです。



Excelの日付と時刻とは

🕒 日付を通し番号で管理する

●日付のシリアル値

Excelの日付は、1900年1月1日を1、翌日を2のように日付ごとに整数の通し番号を割り当てて管理しています。日付に対応する通し番号のことを日付のシリアル値といいます。

・日付と日付のシリアル値の例

1	A	B	C	D	E
2	日付	通し番号			
3	1900/1/1	1			
4	2020/2/28	43889			
5	2020/2/29	43890			
6	2020/3/1	43891			
7	2022/2/28	44620			
8	2022/3/1	44621			
9	2022/6/30	44742			
10	2022/7/1	44743			
11	2022/7/31	44773			

カレンダーは正確です。
月や年によって変化する月末日を
気にする必要はありません。

日付の計算は通し番号を
利用しますが、日付の
表示のままで計算できます。

●日付の入力と認識される日付

数値を「/」（スラッシュ）や「-」（ハイフン）で区切りながら入力すると日付と認識されます。また、年を省略すると今年、日を省略すると1日が補われます。

・セルに入力した日付と認識された日付の例

B8	A	B	C	D
1	セル入力	セル確定後の表示	認識された日付	
2	2022/8/20	2022/8/20	2022/8/20	
3	2022-8-20	2022/8/20	2022/8/20	
4	8/20	8月20日	2022/8/20	
5	8-20	8月20日	2022/8/20	
6	2022年8月20日	2022年8月20日	2022/8/20	
7	令和4年8月20日	令和4年8月20日	2022/8/20	
8	R4.8.20	R4.8.20	2022/8/20	
9	2022年8月	2022年8月	2022/8/1	
10	2022-08	Aug-22	2022/8/1	

入力した日付がどのように
認識されているのかは
数式バーで確認します。

年を省略すると、
今年の年が補われます。

日を省略すると、
1日が補われます。

🕒 24時間を0から1で管理する

●時刻のシリアル値

Excelの時刻は、午前0時を0とし、12時間後の正午で0.5、24時間後の翌日0時で1とする小数で管理しています。この小数を時刻のシリアル値といいます。

・時刻や経過時間と時刻のシリアル値の例

1	A	B	C	D
2	時刻/経過時間	小数		
3	0:00:00	0.00		
4	6:00:00	0.25		
5	12:00:00	0.50		
6	18:00:00	0.75		
7	23:59:59	0.999988426		
8	24:00:00	1.00		
9	36:00:00	1.50		
10	48:00:00	2.00		
11				
12				

24時間経過時の整数部「1」は
日数を表わします。小数部が
時刻のシリアル値であり、
「0」にリセットされます。

次の24時間でさらに1日繰り
上がり、整数部は「2」ですが、
小数部は「0」にリセットされます。

●時刻の入力と認識される時刻

Excelでは、「:」（コロン）、「時」「分」「秒」、「am」や「pm」などを入力すると時刻として認識します。秒を省略すると0秒が補われます。

・セルに入力した時刻と認識された時刻の例

B6	A	B	C	D
1	セル入力	セル確定後の表示	認識された時刻	
2	10時15分	10時15分	10:15:00	
3	10時15分30秒	10時15分30秒	10:15:30	
4	10:15:30	10:15:30	10:15:30	
5	10:15 am	10:15 AM	10:15:00	
6	10:15 pm	10:15 PM	22:15:00	
7				
8				
9				
10				
11				

入力した時刻がどのように
認識されているのかは
数式バーで確認します。

秒を省略すると、
「0」秒が補われます。

なお、日付や時刻の関数では、引数にシリアル値を指定することが多いですが、シリアル値そのものを指定する必要はありません。日付や時刻表示のまま指定します。

日付から曜日を取り出そう

ここで学ぶこと

- WEEKDAY
- 曜日番号
- 種類

日付の曜日番号を求めるには、**WEEKDAY関数**を使います。WEEKDAY関数は月曜日を1、火曜日を2というように、曜日ごとに1～7、もしくは0～6の連続番号を割り当てます。

練習 ▶ 39_商品A売上表、39_パートタイム明細表

書式

WEEKDAY(シリアル値[,種類])



重要用語

WEEKDAY

日付の曜日に対応する曜日番号を求めます。[種類]を指定することで、曜日番号の初期値「1」を何曜日にするのかを決めることができます。通常、1～7の曜日番号ですが、[種類]を「3」にした場合のみ、月曜日を「0」とする0～6の曜日番号となります。また、[種類]を省略すると、「1」を指定したことになり、日曜日始まりの曜日番号になります。



解説

右の表の見方

右の表は、横に日付と曜日、縦にWEEKDAY関数の[種類]を並べています。C列を見ると、2022/5/2は月曜日です。曜日番号が「1」になるのは、[種類]が「2」または「11」のときです。

引数

シリアル値

日付のセルや日付を直接指定します。日付を直接指定するには、日付の前後を「" (ダブルクォーテーション)」で囲みます。たとえば、「"2022/7/16"」のように入力します。「"」で囲まれた日付は日付文字列といえます。

種類

週明けの曜日によって、1、2、3と11～17の数値を指定できます(下図参照)。

=WEEKDAY(B\$2,\$A5)

	A	B	C	D	E	F	G	H	I
1	2022年								
2	日付	5/1	5/2	5/3	5/4	5/5	5/6	5/7	
3	曜日	日	月	火	水	木	金	土	
4	種類	曜日番号							
5	1	1	2	3	4	5	6	7	
6	2	7	1	2	3	4	5	6	
7	3	6	0	1	2	3	4	5	
8	11	7	1	2	3	4	5	6	
9	12	6	7	1	2	3	4	5	
10	13	5	6	7	1	2	3	4	
11	14	4	5	6	7	1	2	3	
12	15	3	4	5	6	7	1	2	
13	16	2	3	4	5	6	7	1	
14	17	1	2	3	4	5	6	7	
15									

1 日付に対応する曜日を番号で表示する

WEEKDAY



解説

曜日番号を求める

ここでは、売上表の日付に対応する曜日番号を求めます。WEEKDAY関数の[種類]を「2」にすると、月曜～金曜の曜日番号は1～5になり、土日は6、7になります。「<=5」(5以下)と「>5」(5より大きい)とすることで、平日と土日に分けられます。SUMIF関数(92ページ参照)で合計を求める条件に曜日番号を利用しています。

=WEEKDAY(A3,2)

① ②

=SUMIF(B:B,H3,E:E)

B列の曜日番号から「<=5」を検索し、E列の売上金額を合計しています。

	A	B	C	D	E	F	G	H	I
1	▼商品A売上表					▼売上分析			
2	日付	曜日番号	販売価格	数量	売上金額	曜日	曜日番号	売上金額	
3	2022/4/1	5	198	30	5,940	月-金	<=5	848,796	
4	2022/4/2	6	224	10	2,240	土日	>5	286,076	
5	2022/4/5	2	198	49	9,702				
6	2022/4/6	3	224	18	4,032				
7	2022/4/7	4	158	50	7,900				

- ① 日付の入ったセル[A3]を[シリアル値]に指定します。
- ② 月曜日から始まる曜日番号にするため、[種類]に「2」を指定します。

2 平日と土日で異なる時給を表示する

WEEKDAY/IF



解説

平日と土日で時給を分ける

ここでは、WEEKDAY関数とIF関数を使って、平日は時給950円、土日は1200円と表示します。

=WEEKDAY(A7,2)

① ②

=IF(B7>=6,\$C\$4,\$C\$3)

③ ④

	A	B	C	D	E	F	G	H	I
1	パートタイム明細表					2022年9月分			
2	氏名	江原 雅美							
3	時給	平日	¥950						
4		土日	¥1,200						
5									
6	日	曜日番号	適用時給	勤務時間	金額				
7	9/1	6	1,200	4.75	¥5,700				
8	9/12	3	950	5.5	¥5,225				
9	9/18	2	950	6.25	¥5,938				
10	9/22	6	1,200	6.25	¥7,500				
11	9/26	3	950	6.75	¥6,413				
12	合計				¥30,775				

- ① 日付を入力したセル[A7]を[シリアル値]に指定します。
- ② 月曜日を1とする曜日番号を割り当てるため、[種類]に「2」を指定します。
- ③ IF関数の[論理式]に「B7>=6」と指定し、曜日番号が6以上かどうか、つまり、土日かどうかを判定しています。
- ④ 曜日番号が6以上の場合は土日の時給のセル[C4]を、6未満の場合は平日の時給のセル[C3]を絶対参照で指定します。

69

表の行列を入れ替えよう

ここで学ぶこと

- TRANSPOSE
- 行と列の入れ替え
- 配列

3行4列の表を4行3列にするなど、**表の行と列を入れ替える**には**TRANSPOSE関数**を使います。TRANSPOSE関数は表全体をひと塊のデータ、すなわち、配列として扱います。

練習 ▶ 69_5S評価_01、69_5S評価_02、69_5S評価_03

書式

TRANSPOSE (配列)

重要用語

TRANSPOSE

Excelの[形式を選択して貼り付け]の[行/列の入れ替え]機能と同様に、表の行と列を入れ替えます。TRANSPOSE関数は配列数式で入力します。

1 表の列見出しを行見出しとして取り出す

TRANSPOSE

解説

列見出しを行見出しとして取り出す

ここでは、表の列見出しを行見出しとして縦方向に取り出します。Excel 2021とMicrosoft 365は、配列の先頭に関数を入力して[Enter]を押すと、スピル機能が動作して過不足なくデータが表示されます。Excel 2019以前は、結果を表示するセル範囲を選択して関数を入力し、[Ctrl]と[Shift]を押しながら[Enter]を押します。

=TRANSPOSE(B2:F2)

H3		=TRANSPOSE(B2:F2)									
		A	B	C	D	E	F	G	H	I	J
1	5S評価								▼集計表		
2	課	整理	整頓	整頓	整頓	整頓	整頓		5S項目	平均値	
3	営業	2	2	2	3	1	2		整理	3.25	
4	管理	2	4	2	2	3			整頓	3.75	
5	生産	5	4	3	4	2			整頓	3	
6	製造	4	5	4	4	4			整頓	2.75	
7									整頓	2.75	

スピル機能により、必要な分だけ過不足なく戻り値が表示されます。

- Excel 2021とMicrosoft 365では、セル[H3]を指定します。Excel2019以前は、セル範囲[H3:H7]を範囲選択します。
- 列見出しのセル範囲[B2:F2]を[配列]に指定します。

2 表の行と列を入れ替える

TRANSPOSE

解説

行と列を入れ替える

ここでは、6列5行で構成された表の行と列を入れ替え5列6行の表にします。

ヒント

スピルとゴースト

スピルは英語のSPILLに由来し「こぼれる」という意味で、関数を入力したセル以外にも結果がこぼれて表示された状態を指します。こぼれたセルはゴーストといいますが、実体がない表現となるのは、ゴーストのセルをダブルクリックすると何も入っていないためです。

7											
8	課	営業	管理	生産	製造						
9	整理		2	5	4						
10	整頓	2	4	4	5						
11	整頓	3	2	3	4						
12	清掃	1	2	4	4						
13	清潔	2	3	2	4						
14											

何も入っていませんが、ゴーストのセルをコピーして値として貼り付けると戻り値が表示されます。

Excel 2021とMicrosoft 365の場合

A8		=TRANSPOSE(A2:F6)									
		A	B	C	D	E	F	G	H	I	J
1	5S評価										
2	課	整理	整頓	整頓	整頓	整頓	整頓				
3	営業	2	2	2	3	1	2				
4	管理	2	4	2	2	2	3				
5	生産	5	4	3	4	2					
6	製造	4	5	4	4	4					
7											
8	課	営業	管理	生産	製造						
9	整理	2	2	5	4						
10	整頓	2	4	4	5						
11	整頓	3	2	3	4						
12	清掃	1	2	4	4						
13	清潔	2	3	2	4						

セル[A8]以外の戻り値はスピル機能の動作により表示されたゴーストです。

=TRANSPOSE(A2:F6)

- 配列の先頭のセル[A8]に、セル範囲[A2:F6]を[配列]に指定して[Enter]で関数を確定します。必要なデータが過不足なく表示されます。

Excel 2019以前の場合

A8		=TRANSPOSE(A2:F6)									
		A	B	C	D	E	F	G	H	I	J
1	5S評価										
2	課	整理	整頓	整頓	整頓	整頓	整頓				
3	営業	2	2	2	3	1	2				
4	管理	2	4	2	2	2	3				
5	生産	5	4	3	4	2					
6	製造	4	5	4	4	4					
7											
8	課	営業	管理	生産	製造						
9	整理	2	2	5	4						
10	整頓	2	4	4	5						
11	整頓	3	2	3	4						
12	清掃	1	2	4	4						
13	清潔	2	3	2	4						

セル範囲の取り方を誤り、範囲が不足している場合は途中までしか表示されないで注意します。

{=TRANSPOSE(A2:F6)}

- 結果を表示するセル範囲をもれなく選択します。ここでは、セル範囲[A8:E13]を選択し、戻り値の範囲が5列6行になるようにします。
- セル範囲[A2:F6]を[配列]に指定し、[Ctrl]と[Shift]を押しながら[Enter]を押して関数の入力を確定します。

72

重複のないデータにしよう

ここで学ぶこと

- UNIQUE
- 重複しないデータ
- データ抽出

リスト形式のフィールドが**どのようなデータで構成されているか**を知りたい場合は **UNIQUE関数**を利用します。また、飛び飛びに入力されたデータから空白を詰めて表示したい場合にも利用することができます。

練習 ▶ 72_交通費_01、72_交通費_02、72_各地域のデータ

書式

UNIQUE (配列 [、列の比較] [、回数指定])



重要用語

UNIQUE

UNIQUE関数は、ユニークデータを抽出します。ユニークデータとは、重複のないデータです。フィールドのデータを抽出したり、1回しか出現しない値を抽出したりすることができます。Excel 2021とMicrosoft 365で利用できます。

引数

配列

表の見出しを除いたセル範囲を指定します。

列の比較

先頭行に列見出しがあり、縦方向(行方向)にデータが並んでいる表は FALSE を指定するか省略します。先頭列に見出しがあり、横方向(列方向)にデータが並んでいる表は TRUE を指定します。

回数指定

指定した配列の中で、1回だけ出現する値を取り出すときは TRUE、配列内の重複しない個別の値を取り出すときは FALSE を指定するか省略します。

1 費目フィールドのデータを抽出する

UNIQUE



解説

列データを抽出する

ここでは、リスト形式の費用データをもとに、費目フィールドのデータを抽出します。スピル機能が動作するため、関数は1箇所に入力すれば、必要な分だけ戻り値が表示されます。

=UNIQUE(B2:B69)

F2 : X Y fx =UNIQUE(B2:B69)					
1	日付	費目	摘要	金額	費目
2	6月2日	旅費交通費	宿泊代	8,200	旅費交通費
3	6月2日	旅費交通費	電車代	11,500	図書費
4	6月2日	旅費交通費	バス代	260	光熱費

- 1 費目フィールドのセル範囲 [B2:B69] を [配列] に指定します。
- 2 [配列] に指定したセル範囲は、縦方向に並んだデータです。したがって、[列の比較] は FALSE を指定しますが、ここでは省略しています。
- 3 [配列] 内のデータの出現回数によらず、重複しないデータを抽出するため FALSE を指定します。ここでは省略しています。

2 費目と摘要のうち、1回しか出現しないデータを抽出する

UNIQUE



解説

1回だけ出現するデータを抽出する

ここでは、費用データをもとに、費目フィールドと摘要フィールドの組み合わせが1回しか出現しないデータを抽出します。

=UNIQUE(B2:C69,FALSE,TRUE)

F2 : X Y fx =UNIQUE(B2:C69,FALSE,TRUE)					
1	日付	費目	摘要	金額	費目
2	6月2日	旅費交通費	宿泊代	8,200	光熱費
3	6月2日	旅費交通費	電車代	11,500	図書費
4	6月2日	旅費交通費	バス代	260	光熱費
5	6月2日	図書費	書籍代	1,850	光熱費
6	6月4日	光熱費	水道代	3,800	通信費
7	6月4日	図書費	新聞購読料	4,900	図書費
8	6月4日	会議費	お茶代	1,120	消耗品費
9	6月4日	会議費	お菓子代	2,860	

- 1 費目フィールドと摘要フィールドのセル範囲 [B2:C69] を [配列] に指定します。
- 2 [配列] に指定したセル範囲は、縦方向に並んだデータです。したがって、[列の比較] は FALSE を指定します。
- 3 [配列] 内のデータのうち、1回だけ出現したデータを抽出するため、TRUE を指定します。

3 空白を詰めてデータを抽出する

UNIQUE



解説

空白以外のデータを抽出する

ここでは、横方向に飛び飛びに入力されているデータから空白を詰めて地域データだけを抽出します。縦方向のデータでも同様です。

=UNIQUE(2:2,TRUE,TRUE)

A7 : X Y fx =UNIQUE(2:2,TRUE,TRUE)									
1	各地域のデータ	出典：政府統計の総合窓口(e-Stat) (https://www.e-stat.go.jp/) から加工して作成							
2	東京都	神奈川県	埼玉県	千葉県	茨城県				
3	小学校数	図書館数	小学校数	図書館数	小学校数	図書館数	小学校数	図書館数	小学校数
4	1328	398	887	85	814	172	777	144	476
5									
6	▼地域抽出								
7	東京都	神奈川県	埼玉県	千葉県	茨城県	群馬県	栃木県		
8									
9									
10									
11									

- 1 地域データのあったセル範囲として行番号「2」を選択し、[配列] に指定します。2行目全体が選択され [2:2] と表示されます。
- 2 [配列] に指定したセル範囲は、横方向に並んだデータです。したがって、[列の比較] は TRUE を指定します。
- 3 [配列] 内のデータのうち、1回だけ出現したデータを抽出するため、TRUE を指定します。



ヒント

[回数指定] を TRUE にすると空白が抽出されない

空白セルを含むデータから重複のないデータを抽出する場合、[回数指定] を FALSE にすると、空白セルも抽出されます。空白セルの戻り値は0になります。[回数指定] を TRUE にすると、何度も出現する空白セルは抽出されません。

表を検索して値を取り出そう②

ここで学ぶこと

- XLOOKUP
- 値の検索
- ワイルドカード

XLOOKUP関数を使うと、VLOOKUP関数と同様に**表を検索して値を取り出す**ことができます。XLOOKUP関数ではVLOOKUP関数よりも機能が拡張されています。

練習▶74_施設情報_01、74_施設情報_02、74_成績表、74_施設情報_03

書式

XLOOKUP(検索値,検索範囲,戻り範囲[,見つからない場合][,一致モード][,検索モード])



重要用語

XLOOKUP

[一致モード]と[検索モード]に指定した方法で、[検索値]を[検索範囲]で検索し、検索結果に対応する[戻り範囲]の値を取得します。[見つからない場合]を指定すると、検索に該当する値がない場合のエラー表示を回避できます。Excel 2021とMicrosoft 365で利用できます。

引数

検索値

検索値となる数値や文字列、または、検索値の入ったセルを指定します。

検索範囲

[検索値]に指定した値を検索するセル範囲を指定します。

戻り範囲

検索によって取得したいデータのセル範囲を指定します。

見つからない場合

検索した結果、該当する値が見つからない場合に表示する文字列を指定します。省略可能です。

一致モード

[検索値]との一致の程度を指定します(下表を参照)。省略すると完全一致で検索します。

検索モード

[検索値]の検索方法を指定します(下表を参照)。省略すると検索範囲の先頭から検索します。

一致モード

指定する値	内容
0	[検索値]と完全に一致する値を検索します。
-1	[検索値]と完全に一致する値が見つからない場合は次に小さい値を検索します。
1	[検索値]と完全に一致する値が見つからない場合は次に大きい値を検索します。
2	[検索値]にワイルドカードを利用する場合に指定します。

検索モード

指定する値	内容
1	[検索範囲]の先頭から末尾に向かって検索します。
-1	[検索範囲]の末尾から先頭に向かって検索します。
-2	VLOOKUP関数の[検索方法]に「TRUE」を指定した場合と同様です。あらかじめ、[検索範囲]を基準に昇順に並べ替えておく必要があります。
2	「-2」の場合と同様ですが、[検索範囲]を基準に降順に並べ替えておく必要があります。

1 指定した施設種別の施設情報を検索する

XLOOKUP



解説

検索値に完全一致するデータを検索する

ここでは、施設情報をもとに、指定した施設種別に一致する施設情報を検索します。[戻り範囲]には複数列の範囲を指定できるため、検索結果に一致するレコードの取得が可能です。

施設種別が「談話室」の場合

=XLOOKUP(E3,A3:A8,A3:C8,"該当なし",0,1)

E7	▼	:	×	✓	<i>fx</i>	=XLOOKUP(E3,A3:A8,A3:C8,"該当なし",0,1)	
	A	B	C	D	E	F	G
1	施設情報			▼検索値			
2	施設種別	定員	備品	施設種別	人数	必要な用品	
3	会議室	12	長机、椅子、ホワイトボード	談話室			
4	会議室	30	長机、椅子、黒板				
5	談話室	10	TVモニター、ビデオ	▼検索結果			
6	和室	6	長机、座布団	施設種別	定員	備品	
7	和室	20	座布団、茶道具	談話室	10	TVモニター、ビデオ	
8	ホール	80	ピアノ、椅子、ボール等				
9							

スピル機能により、必要な分だけ過不足なく戻り値が表示されます。

- セル[E3]を[検索値]に指定します。「談話室」のように[検索値]に直接指定することもできます。
- [検索値]を検索するセル範囲[A3:A8]を[検索範囲]に指定します。
- セル範囲[A3:C8]を[戻り範囲]に指定します。
- [見つからない場合]に「該当なし」と指定します。
- [検索値]に完全一致する値を[検索範囲]の先頭から検索するため、[一致モード]に「0」、[検索モード]に「1」を指定します。

施設種別が「音楽室」の場合

=XLOOKUP(E3,A3:A8,A3:C8,"該当なし",0,1)

E7		:	×	✓	$\sqrt{x}$	=XLOOKUP(E3,A3:A8,A3:C8,"該当なし",0,1)		
	A	B		C	D	E	F	G
1	施設情報				▼検索値			
2	施設種別	定員	備品		施設種別	人数	必要な用品	
3	会議室	12	長机、椅子、ホワイトボード		音楽室	1		
4	会議室	30	長机、椅子、黒板					
5	談話室	10	TVモニター、ビデオ		▼検索結果			
6	和室	6	長机、座布団		施設種別	定員	備品	
7	和室	20	座布団、茶道具		該当なし			
8	ホール	80	ピアノ、椅子、ボール等			2		
9								



解説

エラー表示を回避する

[見つからない場合]を指定すると、検索に該当する値がない場合に表示される[#N/A]エラーの表示を回避することができます。

- 関数を変更しません。セル[E3]に「音楽室」と入力します。
- [検索範囲]に指定した施設種別のセル範囲[A3:A8]に「音楽室」は存在しないため、[見つからない場合]に指定した値が表示されず。